3 (d) Optimisation with McStas:

Finding best parameters to optimise an instrument

E. Farhi <farhi@ill.eu> - Thursday Sept 19th 2013.

References:

Refer to the Workshops and Conferences McStas page at <http://mcstas.org/workshop/>

Scope:

In this exercise, we shall demonstrate how to optimise some parameters of an instrument, that is those that e.g. maximize some criteria. We shall use a SANS instrument with colloidal suspension sample, and study the effect of gravitation. The optimal collimation will be found using parameter scanning, and a full automatic optimisation methodology.

For this exercise, you will need to have iFit installed, from <http://ifit.mccode.org> and select the standalone version (does not require Matlab license).

3D.1: Use a SANS instrument

We shall start from the template SANS instrument, and slightly modify it.



1. Open McGUI and select the Neutron Site/Templates/templateSANS
2. Modify the *Ldetector* component to be a Monitor_nD with automatic limits: options="auto, lambda", bins=1000
3. Run the instrument with lambda=6 Angs neutrons and plot the results. Use log scale (press L-key). Look at the centre of the scattering pattern (X0,Y0). Is it centred ?
4. Would gravitation effect over 3 m distance for 10 Angs neutrons (v=629.62*2π/λ=263 m/s) be visible ?

3D.2: Optimizing a slit position

1. To avoid any direct beam pass through, and not be blocked by the beam stop at the detector, add a perfect beam stop after the AT token on the sample:
   ```
   EXTEND %{
      if (!SCATTERED) ABSORB;
    %}
   ```

2. Add two instrument parameters X0=0 and Y0=0, and modify the 2nd slit position as AT(X0,Y0,6)
3. Save the instrument as *slitSANS.instr* and run it to check that it compiles and executes correctly.
4. Start iFit and 'cd' to the location of the *slitSANS.instr* description.
5. iFit optimisation for McStas is documented on <http://ifit.mccode.org/McStas.html>
6. Many optimizers are available in iFit. Read <http://ifit.mccode.org/Optimizers.html>
7. We shall maximize the 'PSD.dat' total intensity file ('detector' instance) by varying the (X0,Y0) parameters. The X0 and Y0 parameters will be searched in a 6x6 cm² area, starting from X0=-0.01 and Y0=-0.01. To do that, define a structure 'p' holding parameters X0 and Y0:

```
iFit>> p.X0=[-0.02 -0.01 0.02];
p.Y0=p.X0;
```



8. Launch an optimisation with syntax (takes about 100 iterations):

```
[parameters, monitors, status, output]=mcstas('slitSANS.instr', p,
      'monitors=detector; ncount=1e6; mode=optimize; optimizer=fmincmaes');
```

9. Does the optimized slit position fit the theoretical one ?

Solutions to 3D.2:

```
/*******************************************************************************
*
* McStas, neutron ray-tracing package
*         Copyright (C) 1997-2008, All rights reserved
*         Risoe National Laboratory, Roskilde, Denmark
*         Institut Laue Langevin, Grenoble, France
*
* Instrument: slitSANS
*
* %Identification
* Written by: Luke Skywalker
* Date: 19th Dec 2003.
* Origin: The Moon
* Release: McStas CVS_080624
* Version: $Revision: 4382 $
* %INSTRUMENT_SITE: Death_Star
*
* Test instrument for the Sans_spheres component. No guide / velocity selector
* etc.
*
* %Description
* Very simple test instrument for the Sans_spheres component
*
* %Example: lambda=6 Detector: detector_I=6.55e-17
*
* %Parameters
* INPUT PARAMETERS:
* lambda:    Mean wavelength of neutrons      [AA]
* dlambda:   Wavelength spread of neutrons    [AA]
* r:         Radius of scattering hard spheres [AA]
* PHI:       Particle volume fraction [1]
* Delta_Rho: Excess scattering length density (fm/AA^3)
* sigma_abs: Absorption cross section at 2200 m/s [barns]
* X0:        2nd slit horizontal shift [m]
* Y0:        2nd slit vertical shift [m]
*
* %Link
* *
* %End
*******************************************************************************/
DEFINE INSTRUMENT slitSANS(lambda=6, dlambda=0.05, r=100,
  PHI=1e-3, Delta_Rho=0.6, sigma_abs=0.5, X0=0,Y0=0)

TRACE

COMPONENT a1 = Progress_bar()
  AT (0,0,0) ABSOLUTE

COMPONENT arm = Arm(
    )
  AT (0, 0, 0) ABSOLUTE

COMPONENT source = Source_simple(
    radius = 0.02, dist = 3, focus_xw = 0.01, focus_yh = 0.01,
    lambda0 = lambda, dlambda = dlambda, flux = 1e8)
  AT (0, 0, 0) RELATIVE arm

COMPONENT coll1 = Slit(
    radius = 0.005)
  AT (0, 0, 3) RELATIVE arm

COMPONENT coll2 = Slit(
    radius = 0.005)
  AT (X0,Y0, 6) RELATIVE arm

SPLIT COMPONENT sample = Sans_spheres(
  R=r, Phi=PHI, Delta_rho=Delta_Rho, sigma_abs=sigma_abs,
  xwidth=0.01, yheight=0.01, zdepth=0.005)
   AT (0,0,0.2) RELATIVE coll2
   EXTEND %{
     if (!SCATTERED) ABSORB;
   %}

COMPONENT STOP = Beamstop(
    radius = 0.02)
  AT (0, 0, 2.9) RELATIVE sample

COMPONENT detector = PSD_monitor(
    nx = 128, ny = 128, filename = "PSD.dat", xmin = -0.3,
    xmax = 0.3, ymin = -0.3, ymax = 0.3)
  AT (0, 0, 3) RELATIVE sample

COMPONENT PSDrad = PSD_monitor_rad(
    filename = "psd2.dat", filename_av = "psd2_av.dat", rmax = 0.3)
  AT (0, 0, 3.02) RELATIVE sample

END
```